

ANALIZE REPETITIVE. MODELAREA DISPOZITIVELOR ȘI CIRCUITELOR**1. Analize repetitive (parametrice)**

Comenzile .TEMP și .STEP determină repetarea analizelor standard pentru toate valorile unei variabile (temperatură, sursă independentă, parametru global, parametru de model).

.TEMP Comanda de stabilire a temperaturii

Forma generală a comenzii este:

.TEMP <val_temp>*

Exemple:

.TEMP 20 - determină efectuarea analizelor la temperatura de 20°C.

.TEMP -25 0 25 - determină rularea repetată a analizelor la temperaturile de -25°C, 0°C, 25°C.

Comanda .TEMP stabilește temperatura (în °C) la care se efectuează toate analizele din fișierul circuitului. Dacă sunt specificate mai multe temperaturi, toate analizele vor fi rulate pentru fiecare valoare a temperaturii.

Dacă din fișierul circuitului lipsește comanda .TEMP, analizele se efectuează la temperatura nominală (care are valoarea implicită de 27°C sau poate fi stabilită cu parametrul TNOM în comanda .OPTIONS).

.STEP Analiza parametrică

Forme generale ale comenzii sunt:

.STEP [LIN] <nume_var> <val_start> <val_finală> <pas_increm>

.STEP [DEC | OCT] <nume_var> <val_start> <val_finală> <nr_puncte>

.STEP <nume_var> LIST <valoare>*

Exemple:

1) .STEP VCC 0V 12V 1V - determină repetarea analizelor pentru toate valorile sursei VCC, variate liniar de la 0V la 12V, cu pasul de incrementare de 1V;

2) .STEP LIN Iin 10mA -5mA 1mA - determină repetarea analizelor pentru toate valorile sursei Iin, variate liniar de la 10mA la -5mA cu pasul de incrementare de 1mA;

3) .STEP RES RMOD(R) 0.5 1.5 0.1 - determină repetarea analizelor pentru toate valorile parametrului de model R al modelului cu numele RMOD de tip RES, variat liniar de la 0.5 la 1.5, cu pasul de incrementare de 0.1;

4) .STEP DEC NPN QNPN(IS) 1E-18 1E-14 5 - determină repetarea analizelor pentru toate valorile parametrului de model IS al modelului cu numele QNPN de tip NPN, variat logaritm pe decade de la 1E-18A la 1E-14A, cu 5 puncte de calcul pe decadă;

- 5) .STEP TEMP LIST -50 -25 0 25 50 75 - determină repetarea analizelor pentru toate valorile temperaturii (în °C) precizate după cuvântul cheie LIST;
- 6) .STEP PARAM ALPHA 5 10 0.5 - determină repetarea analizelor pentru toate valorile parametrului global ALPHA, variat liniar de la 5 la 10, cu pasul de incrementare de 0.5.

Comanda .STEP determină efectuarea repetată a analizelor standard pentru toate valorile unei variabile, precizate în câmpul <nume_var>.

Comanda .STEP are formele generale de descriere similare celor ale comenzii .DC. Dacă analiza .DC determină calculul repetat al punctului static de funcționare al circuitului, comanda .STEP determină repetarea analizelor standard.

Prima formă generală corespunde variației liniare a variabilei, de la <val_start> la <val_finală>, cu pasul de incrementare <pas_increm>. Cuvântul cheie LIN este opțional. Această formă este utilizată în exemplele 1, 2, 3 și 6.

A doua formă generală corespunde variației logaritmice a variabilei, de la <val_start> la <val_finală>. Scala logaritmică este pe decade dacă în comandă se introduce cuvântul cheie DEC, sau pe octave dacă în comandă se introduce cuvântul cheie OCT. Este obligatoriu ca în comandă să existe unul din cuvintele DEC sau OCT. <nr_puncte> reprezintă numărul de puncte de calcul pe decadă sau octavă. Această formă este utilizată în exemplul 4.

A treia formă corespunde variației variabilei conform unei liste de valori. Această formă este utilizată în exemplul 5.

În primele două forme ale comenzii .STEP, <val_start> poate fi mai mare sau mai mică decât <val_finală>, aceasta înseamnând că variația se poate face în orice direcție. Este obligatoriu însă ca <pas_increm> și <nr_puncte> să fie pozitive.

<nume_var> are aceeași semnificație ca la comanda .DC și poate fi:

- Sursă independentă - nume de sursă independentă de tensiune sau curent (ex. 1 și 2);
- Parametru de model - se specifică tipul de model, numele de model, urmat între paranteze de numele parametrului de model (ex. 3 și 4);
- Temperatură - se specifică cu ajutorul cuvântului cheie TEMP (ex. 5);
- Parametru global - se specifică cuvântul cheie PARAM, urmat de numele parametrului global (ex. 6).

Rezultatele acestei analize pot fi reprezentate grafic, prin intermediul comenzii .PROBE, sau pot fi tipărite în fișierul de ieșire prin intermediul comenzilor .PRINT sau .PLOT.

Pentru o mai bună înțelegere, se vor prezenta două cazuri în care se dorește efectuarea unei analize standard, pentru toate valorile rezistenței rezistorului R1, variate liniar de la 1kΩ la 10 kΩ, cu pasul de incrementare de 1kΩ.

În primul caz se variază un parametru global:

```
R1 1 2 {RVAL}  
.PARAM RVAL=5k  
.STEP PARAM RVAL 1k 10k 1k  
.AC DEC 20 1 1MEG
```

Analiza de semnal mic se va efectua repetat, rezistența rezistorului R1 luând pe rând valorile 1kΩ, 2kΩ, 3kΩ, ... , 10kΩ. Astfel, analiza .AC se va efectua de 10 ori.

În al doilea caz, pentru variația rezistenței rezistorului R1 se variază parametrul de model R (multiplicator de rezistență):

```
R1 1 2 RMOD 5k  
.MODEL RMOD RES(R=1)  
.STEP RES RMOD(R) 1 10 1  
.AC DEC 20 1 1MEG
```

RMOD este numele modelului, iar RES reprezintă tipul dispozitivului. Parametrul de model R este variat liniar de la 1 la 10, cu pasul de incrementare 1, astfel încât valoarea rezistenței rezistorului R1 ia pe rând valorile 1k Ω , 2k Ω , 3k Ω , ... , 10k Ω .

2. Comenzi pentru modelarea dispozitivelor

Comenzile utilizate pentru modelarea dispozitivelor sunt: .MODEL, .DISTRIBUTION, .SUBCKT și .ENDS.

.MODEL Definire model

Forma generală a comenzii este:

```
.MODEL <nume_model> [AKO:<nume_model_ref>] <tip_model>
+([<nume_param>=<valoare> [toleranță]])*
```

Example:

- 1) .MODEL QDEF NPN - definește modelul QDEF al unui tranzistor bipolar NPN, în care parametrii de model au valorile implicite;
- 2) .MODEL RMAX RES(R=0.5 TC1=0.02 TC2=0.005) - definește modelul RMAX al unui rezistor;
- 3) .MODEL DNOM D(IS=1n N=1.5) - definește modelul DNOM al unei diode;
- 4) .MODEL QDRIV NPN(IS=1E-7 BF=50 RB=10) - definește modelul QDRIV al unui tranzistor bipolar NPN;
- 5) .MODEL M1 NMOS(LEVEL=1 VT0=0.7 CJ=0.02pF) - definește modelul M1 al unui tranzistor TEC-MOS cu canal N;
- 6) .MODEL CMOD CAP(C=1 DEV=10%) - definește modelul CMOD al unui condensator. Se specifică valoarea deviației individuale a parametrului de model C al dispozitivului;
- 7) .MODEL DLOAD D(IS=1N DEV=0.5% LOT=10% VJ=0.65V) - definește modelul DLOAD al unei diode. Se specifică valoarea deviației individuale a parametrului de model IS și valoarea deviației de lot a parametrului de model IS;
- 8) .MODEL RMOD RES(R=1 DEV/GAUSS 1% LOT/UNIFORM 5%) - definește modelul RMOD al unei rezistențe. Se specifică deviația individuală a parametrului de model R, bazată pe o distribuție Gaussiană, de valoare 1% și deviația de lot a parametrului de model R, bazată pe o distribuție uniformă, de valoare 5%;
- 9) .MODEL QDRIV2 AKO:QDRIV NPN(BF=40 IKF=50m) - se definește modelul QDRIV2 al unui tranzistor bipolar NPN. Acestui model i se transferă și parametrii de model ai modelului QDRIV din exemplul 4.

Comanda .MODEL definește un set de parametri de model, ce poate fi utilizat de unul sau mai multe dispozitive de același tip.

<nume_model> reprezintă numele atribuit modelului și trebuie să înceapă cu o literă (se recomandă utilizarea literei rezervate descrierii dispozitivului respectiv).

Opțional, cu ajutorul cuvântului cheie AKO, se pot transfera parametrii de model ai unui model de referință precizat cu <nume_model_ref> la modelul curent. Se vor transfera doar parametrii de model care apar în modelul de referință și nu apar în modelul curent, însă fără toleranțe. În exemplul 9, modelului QDRIV2 i se transferă parametrii modelului QDRIV din exemplul 4. Astfel, modelul QDRIV2 are parametrii IS și RB cu valorile specificate în

modelul QDRIV, iar parametrii BF și IKF au valorile specificate în modelul curent (valoarea parametrului BF nu se transferă deoarece este specificată în modelul curent).

<tip_model> specifică tipul dispozitivului pentru care se definește modelul. Tipul modelului se precizează cu unul din cuvintele cheie:

CAP - condensator	IND - bobină
RES - rezistor	D - diodă
NPN - TB NPN	PNP - TB PNP
LPNP - TB PNP lateral	NJF - TEC-J cu canal N
PJF - TEC-J cu canal P	NMOS - TEC-MOS cu canal N
PMOS - TEC-MOS cu canal P	GASFET - TEC-MES GaAs cu canal N
CORE - miez magnetic neliniar	VSWITCH - comutator controlat în tensiune
ISWITCH - comutator controlat în curent	TRN - linie de transmisie cu pierderi

După <tip_model> urmează lista parametrilor de model ai dispozitivului specificat. Pot fi specificați toți, oricare sau nici unul din parametrii de model. Parametrii neinițializați utilizează valorile lor implicite. Fiecărui parametru i se poate specifica o toleranță, în formatul:

[DEV[/<nume_distribuție>] <valoare>[%]] [LOT[/<nume_distribuție>] <valoare>[%]]

Toleranța este luată în considerație în cadrul analizei statistice .MC, în care analizele standard se efectuează repetat, la fiecare repetare parametrii de model luând valori aleatoare în jurul valorii nominale, în interiorul deviației specificate. Dacă după DEV sau LOT se specifică <nume_distribuție>, trebuie să existe, fără spațiu, caracterul ”/”.

DEV și LOT reprezintă tipul deviațiilor valorilor parametrilor de model (individuale ale dispozitivului, respectiv de lot de dispozitive). Dacă unui parametru de model i se specifică numai valoarea deviației de lot, în cadrul unei rulări, toate dispozitivele care utilizează modelul respectiv vor avea aceeași valoare pentru parametrul respectiv. Dacă unui parametru de model i se specifică numai valoarea deviației individuale, atunci parametrul respectiv ia valori aleatoare pentru fiecare dispozitiv care utilizează acest model.

<nume_distribuție> și distribuția implicită (setată cu parametrul DISTRIBUTION al analizei .OPTIONS) pot fi următoarele:

UNIFORM - generează deviații distribuite uniform în intervalul $\pm$ <valoare>;

GAUSS - generează deviații distribuite Gaussian în intervalul $\pm 3\sigma$, iar <valoare> specifică deviația $\pm 1\sigma$;

<nume_distribuție_utilizator> - generează deviații conform unei distribuții definite de utilizator, iar <valoare> specifică deviația ± 1 în definiția utilizatorului (a se vedea comanda .DISTRIBUTION).

.DISTRIBUTION Distribuție definită de utilizator


Forma generală a comenzii este:

.DISTRIBUTION <nume_distribuție> (<deviație> <probabilitate>)*

Exemple:

1) .DISTRIBUTION bimodal (-1,1) (-0.5,1) (-0.5,0) (0.5,0) (0.5,1) (1,1) - definește distribuția din figura din stânga jos;

2) .DISTRIBUTION triunghi $(-1,0)$ $(0,1)$ $(1,0)$ - definește distribuția din figura din dreapta jos;



Comanda .DISTRIBUTION definește o distribuție de probabilitate cu numele <nume_distribuție>, utilizată în descrierea toleranțelor parametrilor de model ai dispozitivelor. Toleranțele sunt utile în cazul analizelor statistice (.MC și .WCASE).

Curba descrisă în comanda .DISTRIBUTION controlează distribuția relativă de probabilitate a numerelor generate aleator de PSPICE pentru calculul deviațiilor parametrilor de model. Curba distribuției este obținută prin interpolarea liniară a punctelor (<deviație> <probabilitate>). Fiecare <deviație> trebuie să aparțină intervalului $(-1,1)$ și trebuie introdusă în listă în ordine crescătoare. Fiecare <probabilitate> reprezintă o probabilitate relativă și trebuie să fie mai mare sau egală cu zero.

În cadrul analizelor statistice etapele în care se reactualizează deviația valorii unui parametru sunt următoarele:

1. PSPICE generează un număr aleator temporar în intervalul $(0,1)$;
2. Se normalizează aria de sub distribuția specificată;
3. Se stabilește numărul aleator final la valoarea în care aria de sub distribuția normalizată este egală cu numărul aleator temporar;
4. Numărul aleator final se înmulțește cu toleranța specificată.

.SUBCKT Definiție de subcircuit
.ENDS Sfârșit definiție de subcircuit

Forma generală a descrierii unui subcircuit este:

```
.SUBCKT <nume_subcircuit> [noduri_subcircuit]*
+[PARAMS: <<nume_param_local>=<valoare>>]*
<descriere dispozitive>*
[comenzi .IC, .NODESET, .MODEL, .PARAM, .FUNC]*
.ENDS
```

Exemple:

```
.SUBCKT AMP_OP 1 2 100 101 17
...
.ENDS

.SUBCKT FILTRU IN IES PARAMS: F_CENTRALA=100kHz BANDA=10kHz
...
.ENDS
```

Comanda .SUBCKT indică începutul definiției unui subcircuit, iar sfârșitul definiției lui este marcat prin intermediul comenzii .ENDS. Toate declarațiile cuprinse între .SUBCKT și .ENDS constituie definiția subcircuitului.

Apelarea unui subcircuit se face prin intermediul unui dispozitiv X, iar PSPICE include în circuit toate dispozitivele din descrierea subcircuitului.

Parametrii comenzii .SUBCKT au următoarea semnificație:

<nume_subcircuit> reprezintă numele subcircuitului și este utilizat de un dispozitiv X (apelare subcircuit);

[noduri_subcircuit]* este o listă opțională de pini ai subcircuitului, care se vor conecta la nodurile circuitului principal. În această listă de noduri nu este permisă utilizarea etichetei 0 (zero), deoarece acesta este rezervată pentru nodul de masă globală a circuitului. Numărul de noduri din declarația de apelare a subcircuitului trebuie să fie egal cu numărul de noduri din definiția subcircuitului.

Apelarea unui subcircuit în circuitul principal se face prin intermediul unui dispozitiv, a cărui formă generală de descriere este:

```
X<nume> [noduri_cp]* <nume_subcircuit>
+[PARAMS: <<nume_param_local>=<valoare>>*]
```

Parametrii apelării subcircuitului au următoarea semnificație:

[noduri_cp]* reprezintă lista nodurilor din circuitul principal la care se vor conecta, în ordine, pinii subcircuitului, specificați în lista [noduri_subcircuit]* ai comenzii .SUBCKT.

<nume_subcircuit> reprezintă numele subcircuitului apelat.

Specificația [PARAMS: <<nume_param_global>=<valoare>>*] este opțională și permite utilizarea în subcircuit a unor parametri locali. În exemplul 2 de mai sus, pentru subcircuitul FILTRU se definesc parametrii locali F_CENTRALA=100KHz și BANDA=10KHz.

Pentru o mai bună înțelegere a celor prezentate anterior, consideram două exemple:

Exemplul 1:

```
.SUBCKT AO 1 2 10 11 14
...
.ENDS
X1 2 6 50 51 7 AO
X2 0 13 50 51 27 AO
```

Subcircuitul AO este apelat în circuitul principal de două ori. La prima apelare (prin intermediul dispozitivului X1), nodurile 1, 2, 10, 11 și 14 ale subcircuitului se conectează la nodurile 2, 6, 50, 51 și 7 din circuitul principal. La a doua apelare (prin intermediul dispozitivului X2), nodurile 1, 2, 10, 11 și 14 ale subcircuitului se conectează la nodurile 0, 13, 50, 51 și 27 din circuitul principal.

Exemplul 2:

```
.SUBCKT ETAJ_RC 1 3 PARAMS: RVAL=3k
R1 1 2 {RVAL}
R2 2 3 {2*RVAL}
C1 1 0 100N
C2 2 0 50N
.ENDS
```

```
X1 5 6 ETAJ_RC PARAMS: RVAL=5k
X2 7 8 ETAJ_RC
```

În acest caz, numele subcircuitului este ETAJ_RC, iar pinii lui sunt etichetați cu 1 și 3. La definirea subcircuitului este declarat parametrul local RVAL, căruia i se atribuie valoare implicită de 3k, aceasta transferându-se în interiorul subcircuitului, în expresiile în care intervine RVAL.

Subcircuitul este apelat de două ori în circuitul principal. În cazul apelării subcircuitului prin intermediul dispozitivului X1, terminalele 1 și 3 ale subcircuitului sunt conectate la nodurile 5, respectiv 6 din circuitul principal, iar parametrul local RVAL i se redefineste valoarea, adică 5k.

În cazul apelării subcircuitului prin intermediul dispozitivului X2, terminalele 1 și 3 ale subcircuitului sunt conectate la nodurile 7, respectiv 8 din circuitul principal, iar parametrul local RVAL are valoarea implicită, adică 3k.


Observații: Nodurile, dispozitivele și numele de model sunt locale în subcircuitul în care sunt definite. Acest lucru permite ca în subcircuit să se poată utiliza un nume (de nod, dispozitiv, model) care se utilizează și în circuitul principal. După expandarea subcircuitului în circuitul principal, toate numele dispozitivelor din subcircuit au ca prefix numele utilizat pentru apelarea subcircuitului respectiv. De exemplu, Q8 devine X2.Q8, iar nodul 5 devine X2.5. Ca urmare, după expandare, toate numele vor deveni unice.

Subcircuitul pot fi înlanțuite, adică între comenzile .SUBCKT și .ENDS pot fi descrise dispozitive de apelare ale altor subcircuite (X), însă nu este permisă descrierea unui alt subcircuit (o altă pereche de comenzi .SUBCKT și .ENDS).

3. Aplicații

Aplicația 1:


Fie circuitul următor:




Filtrul RLC se va descrie ca subcircuit, în care se vor defini parametrii locali RVAL, LVAL, CVAL cu valorile de 5k, 100u, 10n. Acest subcircuit va fi apelat de 3 ori. Efectuați o analiză de semnal mic, în intervalul (10Hz, 100MHz) și reprezentați grafic modulul funcțiilor de transfer ale celor 3 filtre.

Aplicatia 2:

Fie circuitul derivator în domeniul timp, pentru care $v_{IES}(t) = -k \frac{v_{IN}(t)}{dt}$:



Sursa V1 generează forma de undă de mai jos:



Dispozitivul X1 este un amplificator operațional UA741, descris ca subcircuit în fișierul bibliotecă de modele OPAMP.LIB.

Efectuați analiza în domeniul timp a circuitului, cu parametrii .TRAN 1u 5m 0 100u și reprezentați grafic intrarea și ieșirea circuitului.